SOLVING DIFFERENTIAL EQUATIONS ANALYTICALLY WITH MAPLE'S ASSISTANCE

Karen Donnelly
Saint Joseph's College
Department of Mathematics
Highway 231
Rensselaer, IN 47978
karend@saintjoe.edu

A "by hand" analytic solution to a differential equation can require multiple pages and has a high probability of computational error. Using Maple™ to handle messy manipulations, students are able to complete correct solutions in a reasonable time, focus on methods, and enjoy their work. At Saint Joseph's College we use Maple throughout the Differential Equations course to assist in computations and visualization. Two examples are illustrated below: the method of undermined coefficients and the method of Laplace transforms. Maple's symbolic computation facilities are very helpful in supporting problem solving using these methods. For Laplace transform exercises, students can provide elegant, yet painless, step by step solutions using Maple's built-in Laplace and inverse Laplace transform, Heaviside, and partial fraction conversion functions. Students can then use Maple's **dsolve** command to verify the correctness of their solutions. Graphing their solutions using **plot** or **DEplot** provides additional insight.

1. The Method of Undetermined Coefficients

This method can be applied a differential equation

$$a\frac{d^2y}{dt^2}+b\frac{dy}{dt}+cy=f(t)$$

where $f(t)$ is of a particular form. To solve, we must:

1. Find the general solution to the homogeneous equation.
2. Find a particular solution to the nonhomogeneous equation. To do this we:
    a) Make the appropriate guess.
    b) Substitute the guess into the differential equation.
    c) Solve for the unknown coefficients in the guess.
    d) Substitute these found coefficients into the guess.
3. Add the homogeneous and the particular solutions for a general solution.

The computationally messy, error-prone part of this process is finding a particular solution, where Maple can help as is illustrated below for the differential equation

$$\frac{d^2y}{dt^2}+y=4t\cos(t).$$

We first define the differential equation (ODE).

```
> ODE := diff( y(t), t$2 ) +  y(t) = 4*t*cos(t):
```

We then proceed with the following steps to find a particular solution to the nonhomogeneous differential equation.

a) Make a guess.

```
> GUESS :=t*(A*t+B)*cos(t) + t*(C*t+D)*sin(t):
```

b) Substitute the guess into the differential equation and evaluate.

```
> GOAL :=subs( y(t)=GUESS, ODE): GOAL := eval(GOAL);
```

$$GOAL := 2\,A\cos(t) - 2\,(A\,t + B)\sin(t) - 2\,t\,A\sin(t) + 2\,C\sin(t) + 2\,(C\,t + \mathrm{D})\cos(t) + 2\,t\,C\cos(t) = 4\,t\cos(t)$$

c) Solve for coefficients. Gather the coefficients of like terms, set the coefficients of each term equal to 0, and solve:

```
> GOAL := lhs(GOAL)-rhs(GOAL)=0:
> collect(GOAL,[sin(t),cos(t),t]);
```

$$(2\,C - 4\,A\,t - 2\,B)\sin(t) + ((4\,C - 4)\,t + 2\,A + 2\,\mathrm{D})\cos(t) = 0$$

```
> EQNS := { -4*A = 0, 2*C-2*B=0,4*C-4=0,2*A + 2*D = 0}:
> COEFFS := solve( EQNS, {A,B,C,D} );
```

$$COEFFS := \{A = 0, \mathrm{D} = 0, C = 1, B = 1\}$$

d) Substitute the found coefficients into the guess.

```
> pSOLN := subs( COEFFS, GUESS );
```

$$pSOLN := t\cos(t) + t^2\sin(t)$$

Check the answer. (If Maple returns 0 the solution satisfies the differential equation.)

```
> odetest(y(t)=pSOLN,ODE);
```

$$0$$

2. The Method of Laplace Transforms

This method is useful for solving initial value problems involving linear, constant coefficient equations. When using Laplace transforms, derivatives are converted into powers, transforming the differential equation into an algebraic equation that can be solved by simple algebraic techniques. The inverse Laplace transform is then applied to give the corresponding function which is a solution to the differential equation.

To demonstrate the method we consider a mixing problem with valve switching. The mixing tank initially contains 300 g of salt mixed into 1000 L of water. Initially 6 liters of solution with 4 g of salt per liter of solution per minute is entering the tank through valve A. We close valve A and open value B at time t = 10 minutes. The solution then enters at a rate of 6 liters of solution with 2 g of salt per liter. Hence the rate of input of salt changes from 24 g/min to 12 g/min at time t = 10. The solution is kept well stirred in the tank and leaves at a rate of 6 liters per minute. This leads to a differential equation whose right hand side has a jump discontinuity. The differential equation modeling this problem is given by

$$\frac{dy}{dt}+\frac{3}{500}y=\begin{cases} 0, & t\le 0 \\ 24, & 0<t\le 10 \\ 12, & t>10 \end{cases}$$

$$y(0)=300 \text{ grams.}$$

The following initial Maple commands load needed libraries and set up a convenient notation for the Laplace transform and the unit step function.

```
> restart; with(inttrans): with(student):
> alias(Y(s)=laplace(y(t),t,s)):
> _EnvUseHeavisideAsUnitStep := true:
> alias(u = Heaviside):
```

We next define the differential equation to be solved (ODE) along with initial conditions (IC) and have Maple solve directly.

```
> ODE := diff( y(t), t)+(3/500)* y(t) = 24*u(t)
        - 12*u(t-10):
> IC := y(0) = 300:
> dsolve({ODE, IC},y(t),method = laplace);
```

$$y(t)=-3700\mathbf{e}^{\left(-\frac{3}{500}t\right)}+4000-2000u(t-10)\left(1-\mathbf{e}^{\left(-\frac{3}{500}t+\frac{3}{50}\right)}\right)$$

Now our step by step solution with Maple:

Step 1. Apply the Laplace transform to the entire equation.

```
> ODE_lap := laplace( ODE, t, s );
```

$$ODE\_lap := s\,Y(s)-y(0)+\frac{3}{500}Y(s)=\frac{12\,(2-\mathbf{e}^{(-10\,s)})}{s}$$

Step 2: Substitute in the initial conditions.

```
> ODE_lap := subs( IC, ODE_lap );
```

$$ODE\_lap := s\,Y(s) - 300 + \frac{3}{500}\,Y(s) = \frac{12\,(2 - \mathbf{e}^{(-10\,s)})}{s}$$

Step 3: Next, it is necessary to explicitly solve for the Laplace transform of the solution.

```
> SOLN_lap := solve( ODE_lap, { Y(s) } );
```

$$SOLN\_lap := \left\{ Y(s) = \frac{6000(25s + 2 - e^{(-10s)})}{s(500s+3)} \right\}$$

```
> SOLN_lap := expand(SOLN_lap,s);
```

$$SOLN\_lap := \left\{ Y(s) = \frac{150000}{500s+3} + \frac{12000}{s(500s+3)} - \frac{6000}{s(500s+3)(\mathbf{e}^{s})^{10}} \right\}$$

Step 4: Now we have our algebraic solution "Y(s)". We must find the inverse Laplace transform of

$$\frac{150000}{500s+3} + \frac{12000}{s(500s+3)} - \frac{6000}{s(500s+3)e^{10s}} .$$

If solving by hand we use a partial fraction expansion for $\frac{1}{s(500s+3)}$ to calculate the inverse Laplace transform of the second and third terms.

```
> convert(1/(s*(500*s+3)),parfrac,s);
```

$$\frac{1}{3\,s} - \frac{500}{3\,(500\,s+3)}$$

Maple can do all of the work for computing the inverse, against which we can check our by hand solution.

```
> SOLN := invlaplace( SOLN_lap, s, t ):
```

```
> expand(SOLN);
```

$$\left\{ y(t) = -3700\mathbf{e}^{\left(-\frac{3}{500}t\right)} + 4000 - 2000u(t-10) + 2000u(t-10)\mathbf{e}^{\left(-\frac{3}{500}t\right)}\mathbf{e}^{\left(\frac{3}{50}\right)} \right\}$$

It is helpful to plot the solution to see what the graph of the amount of salt in grams as a function of time *t* in minutes.

```
> answer := rhs(SOLN[1]):
> plot(answer, t = 0..15, y =
0..600,color=black,thickness=2);
```

The output of the above plot command clearly indicates the change in the solution y(t) at time t = 10 seconds.

Solution to Mixing Problem



Figure 1

The two examples above show how Maple can be a friendly assistant in the analytic solution of differential equations. Our experience has been that when using this software package, students can complete exercises successfully with less frustration and can concentrate on modeling and methods. Grading of correct homework then becomes easy, and the instructor is not spending hours helping students "debug" their solutions.

The complete Maple worksheets from which the above excerpts are taken can downloaded from http://www.saintjoe.edu/~karend/DEMaple. The URL for the differential equations course is http://www.saintjoe.edu/~karend/m336.

References:

1. Maple Application Center – Maplesoft. <http://www.maplesoft.com/applications>

2. Nagle, R.Kent, Edward B. Saff, and Arthur D. Snider. Fundamentals of Differential Equations. 6th Ed. Addison Wesley, 2004.